# $Id$
# Maintainer : Felix Yan <felixonmars@archlinux.org>
# Contributor: Ionut Biru <ibiru@archlinux.org>
# Contributor: William Rea <sillywilly@gmail.com>

pkgbase=python-chardet
pkgname=('python-chardet' 'python2-chardet')
pkgver=3.0.4
pkgrel=1
arch=('any')
url="https://github.com/chardet/chardet"
license=('LGPL')
makedepends=('python-setuptools' 'python2-setuptools')
source=("https://pypi.io/packages/source/c/chardet/chardet-${pkgver}.tar.gz")
sha512sums=('61a03b23447a2bfe52ceed4dd1b9afdb5784da1933a623776883ee9f297e341f633e27f0ce0230bd5fdc5fdb5382105ab42736a74a417ddeb9f83af57455dba5')

prepare() {
   cp -a chardet-$pkgver{,-py2}
}

package_python-chardet() {
   depends=('python-setuptools')
   pkgdesc="Python3 module for character encoding auto-detection"

   cd chardet-$pkgver
   python setup.py install --root="$pkgdir" --optimize=1
}

package_python2-chardet() {
   depends=('python2-setuptools')
   pkgdesc="Python2 module for character encoding auto-detection"

   cd chardet-$pkgver-py2
   python2 setup.py install --root="$pkgdir" --optimize=1

   # To avoid file conflict with the python3 version
   mv "$pkgdir"/usr/bin/chardetect{,-py2}
}
